* Import ENH 1996-4
* This file converts the files from the ENH 1996-4 into Stata format. 
* Inputs: Files in Data/Original/encuestas de hogares/ENH/1996/4/eh94dc96.P99
* Outputs: Files in Data/Importable/ENH/TRIM 4 

clear all
version 13.1

* Change
project, doinfo   
local master "`r(pdir)'"
local doname "`r(dofile)'" 

cap mkdir "`master'/Data/Importable/ENH/1996/TRIM 4"
cd "../../Data/Original/encuestas de hogares/ENH/1996"
cd eht19964

* Registro tipo 10 - Caracteristicas generales
infix tipo_registro 1-2 str id 3-24 str municipio 3-4 segmento 5-7 estrato 8 sector 9-12 seccion 13-14 manzana 15-16 vivienda 17-19 hogar 20-21 personas_hogar 22-23 resultado_encuesta 24 orden 25-26 ///
parentesco 27-28 sexo 29 edad 30-31 estado_civil 32 str departamento_nacimiento 33-34 str municipio_nacimiento 35-37  cabecera_nacimiento 38 tiempo_residencia 39-40 asiste_escuela 41 colegio_oficial 42 ultimo_grado_estudios 43-45 lee_escribe 46 ///
area 93-94 fex 95-98 ///
if tipo_registro==10 using eh94dc96.P99 , clear

save "`master'/Data/Importable/ENH/1996/TRIM 4/1996_4_general.dta" , replace
project, creates("`master'/Data/Importable/ENH/1996/TRIM 4/1996_4_general.dta")

* Registro tipo 50 - Fuerza de trabajo
clear
infix tipo_registro 1-2 str id 3-24 str municipio 3-4 segmento 5-7 estrato 8 sector 9-12 seccion 13-14 manzana 15-16 vivienda 17-19 hogar 20-21 personas_hogar 22-23 resultado_encuesta 24 orden 25-26 ///
actividad 27 trabajo_familia 28 no_trabajo_empleo 29 razon_ausente 30 busco_ultima 31 ha_buscado 32 no_ha_buscado_t 33-34 d_ing_trabajo 35 d_ing_intereses 36 d_ing_arriendo 37 d_ing_pension 38 d_ing_ayudas 39 d_ing_otros 40 d_ing_no 41 ingreso 42-49 ///
area 93-94 fex 95-98 ///
if tipo_registro==50 using eh94dc96.P99 , clear

save "`master'/Data/Importable/ENH/1996/TRIM 4/1996_4_fuerza_trabajo.dta" , replace
project, creates("`master'/Data/Importable/ENH/1996/TRIM 4/1996_4_fuerza_trabajo.dta")

* Registro tipo 60 - Ocupados
clear
infix tipo_registro 1-2 str id 3-24 str municipio 3-4 segmento 5-7 estrato 8 sector 9-12 seccion 13-14 manzana 15-16 vivienda 17-19 hogar 20-21 personas_hogar 22-23 resultado_encuesta 24 orden 25-26 ///
trabajo_permanente 27 horas_ultima 28-29 horas_semana 30-31 quiere_trabajar_mas 32 motivo_trab_mas 33 ocupacion 34-35 actividad_empresa 36-37 tipo_trabajador 38 salario 39-46 salario_period 47 d_salario_alimentos 48 salario_alimentos 49-56 ganancia_mes 57-64 d_otros_ingresos 65 otros_ingresos 66-73 afiliado_salud 74 afiliado_pension 75  ///
area 93-94 fex 95-98 ///
if tipo_registro==60 using eh94dc96.P99 , clear
save "`master'/Data/Importable/ENH/1996/TRIM 4/1996_4_ocupados.dta" , replace
project, creates("`master'/Data/Importable/ENH/1996/TRIM 4/1996_4_ocupados.dta")

* Registro tipo 70 - Desocupados
clear
infix tipo_registro 1-2 str id 3-24 str municipio 3-4 segmento 5-7 estrato 8 sector 9-12 seccion 13-14 manzana 15-16 vivienda 17-19 hogar 20-21 personas_hogar 22-23 resultado_encuesta 24 orden 25-26 ///
tiempo_buscando 27-28 buscado_pte 29 buscado_tcompleto 30 buscado_tipo 31 buscado_ocupacion 32-33 buscado_actividad 34-35 buscado_primeravez 36 tiempo_dejo_trabajar 37-38 trabajoo_ocupacion 39-40 trabajoo_actividad 41-42 trabajoo_tipo 43 d_ing_trabajo 44 d_ing_intereses 45 d_ing_arriendos 46 d_ing_pensiones 47 d_ing_ayudas 48 d_ing_otros 49 d_ing_no 50 ingreso 51-58 ///
area 93-94 fex 95-98 ///
if tipo_registro==70 using eh94dc96.P99 , clear
save "`master'/Data/Importable/ENH/1996/TRIM 4/1996_4_desocupados.dta" , replace
project, creates("`master'/Data/Importable/ENH/1996/TRIM 4/1996_4_desocupados.dta")

* Registro tipo 80 - Inactivos
clear 
infix tipo_registro 1-2 str id 3-24 str municipio 3-4 segmento 5-7 estrato 8 sector 9-12 seccion 13-14 manzana 15-16 vivienda 17-19 hogar 20-21 personas_hogar 22-23 resultado_encuesta 24 orden 25-26 ///
segunda_actividad 27 tiempo_inactivo 28 actividad_antes 29-30 tiempo_trabajando 31-32 motivo_dejo_trabajo 33 ha_buscado 34 cuando_busco 35-36 motivo_no_buscado 37 ///
area 93-94 fex 95-98 ///
if tipo_registro==80 using eh94dc96.P99 , clear
save "`master'/Data/Importable/ENH/1996/TRIM 4/1996_4_inactivos.dta" , replace
project, creates("`master'/Data/Importable/ENH/1996/TRIM 4/1996_4_inactivos.dta")

* Registro tipo 90 - Salud y Seguridad Social

clear 
infix tipo_registro 1-2 str id 3-24 str municipio 3-4 segmento 5-7 estrato 8 sector 9-12 seccion 13-14 manzana 15-16 vivienda 17-19 hogar 20-21 personas_hogar 22-23 resultado_encuesta 24 orden 25-26 ///
cubierto_ss 27 regimen_salud 28 tipo_entidad 29 str entidad 30-32 cotizante 33 complementario 34 medicina_div 35 seguro_salud 36 seguro_hosp 37 otro 38 ninguno 39 ultimo_mes_complementario 40 ultimo_mes_medicina 41 ultimo_mes_salud_o 42 ultimo_mes_no 43 cons_consulta_externa 44 cons_urgencias 45 cons_control 46 cons_chequeo 47 cons_otro 48 cons_na 49 ser_medgen 50 ser_medesp 51 ser_odogen 52 ser_odoesp 53 ser_labclin 54 ser_hosp 55 ser_cirugia_hosp 56 ser_cirugia_amb 57 ser_rehabilitacion 58 ser_otro 59 ser_na 60 medicamentos 61 pago_medicamentos 62 ///
area 93-94 fex 95-98 ///
if tipo_registro==90 using eh94dc96.P99 , clear
save "`master'/Data/Importable/ENH/1996/TRIM 4/1996_4_saludss.dta" , replace
project, creates("`master'/Data/Importable/ENH/1996/TRIM 4/1996_4_saludss.dta")



